Dynamics of Machines Week 11 – Exercise

Laval Rotor, Critical Rotational Speed of Rotors



Disk of the Laval rotor. Point S is the center of gravity and point O is the geometrical center point of the rotor. The distance between point S and O is the eccentricity (e).

Vibration modes of the Laval rotor



Vibration mode of the Laval rotor in the first critical rotational speed.



Vibration mode of the Laval rotor in the second critical rotational speed.

Parameters of the Laval rotor:

$$\alpha = 60(rad/s)\ ,\ \Omega_{max} = 120(rad/s)\ ,\ e = 0{,}001\ (m)\ ,\ t_{start} = 5\ (s)\ ,\ \xi = 0{,}1$$

<u>Task:</u> Plot the x and y displacement coordinates of the rotor in one diagram! The rotor revs up from rest to the maximal rotational speed during $t_{start} = 5(s)$. The rotor will be worked to $t_{max} = 10(s)$. Length of a time step is $\Delta t = 0{,}001\ (s)$.

Equations of motion

$$\ddot{x} + 2\alpha\xi\dot{x} + \alpha^2 x = e\,\omega^2 \cos(\omega t)$$

$$\ddot{y} + 2\alpha\xi\dot{y} + \alpha^2 y = e\,\omega^2 \sin(\omega t)$$

Initial Conditions: $t_0 = 0\,s \qquad x_0 = 0\ (m)\ ,\ \dot{x}_0 = 0\ (m/s),$

$y_0 = 0\ (m)\ ,\ \dot{y}_0 = 0\ (m/s).$

$$\ddot{x} = -2\alpha\xi\dot{x} - \alpha^2 x + e\,\omega^2 \cos(\omega t) \qquad \text{(Rearrangment)}$$

$$\ddot{y} = -2\alpha\xi\dot{y} - \alpha^2 y + e\,\omega^2 \sin(\omega t) \qquad \text{(Rearrangment)}$$




```
// week 11 – Laval rotor

clear;
usecanvas (%F);

// Input variables:
alfa=60;                 // undamped natural frequency (rad/s)
omega_max=120;           // max. angular velocity of the rotor (rad/s)
exc=0.001;               // eccentricity (m)
tstart=5;                // rev up time of the rotor (s)
tmax=10;                 // final time (s)
omega_dot=omega_max/tstart;          // angular acceleration during rev up period of the rotor
kszi=0.1;                // Lehr damping coefficient

z0=zeros(4,1);           // Initial conditions
t0=0;
dt=0.001;                // length of one time step (s)
t=0:dt:tmax;             // time interval matrix (s)
```

```
// Calculation ------------------------------------------------------------------------------------------------
function zdot=xyrotor(t, z)
        if t < tstart then
              OM=omega_dot*t;
        else OM=omega_max;
        end
  zdot(1)= z(2);                    // x coordinate of the velocity
  zdot(2)=-2*kszi*alfa*z(2)-alfa^2*z(1)+ exc*OM^2*cos(OM*t);          // x coordinate of the acceleration
  zdot(3)= z(4);                    // y coordinate of the velocity
  zdot(4)=-2*kszi*alfa*z(4)-alfa^2*z(3)+ exc*OM^2*sin(OM*t);          // y coordinate of the acceleration
endfunction

z=ode("rk",z0,t0,t,xyrotor);

x=z(1,:);              // x coordinate of the displacement
y=z(3,:);              // y coordinate of the displacement

// Plotting ------------------------------------------------------------------------------------------------
scf(1)
aa=gca();
aa.isoview='on';

plot2d(x,y)
//comet(x,y)

xtitle("Vortex motion of the geometrical center point of the rotor","x [mm]", "y [mm]")
xgrid(2)

scf(2)
plot(t,sqrt(x^2+y^2)*1000,"k")
xtitle("Displacement of the geometrical center point of the rotor","displacement [mm]", "time [s]")
xgrid(2)
```


Vortex motion of the geometrical center point of the rotor
y[mm]
x[mm]
0.0024
0.0022
0.002
0.0018
0.0016
0.0014
0.0012
0.001
0.0008
0.0006
0.0004
0.0002
0
-0.0002
-0.0004
-0.0006
-0.0008
-0.001
-0.0012
-0.0014
-0.0016
-0.0018
-0.002
-0.0022
-0.0024
-0.0015
-0.001
-0.0005
0
0.0005
0.001
0.0015
0.002
0.0025


Displacement of the geometrical center point of the rotor
time [s]
displacement [mm]
2.4
2.3
2.2
2.1
2
1.9
1.8
1.7
1.6
1.5
1.4
1.3
1.2
1.1
1
0.9
0.8
0.7
0.6
0.5
0.4
0.3
0.2
0.1
0
0
0.5
1
1.5
2
2.5
3
3.5
4
4.5
5
5.5
6
6.5
7
7.5
8
8.5
9
9.5
10